

irf ctable — Combined tables of IRFs, dynamic-multiplier functions, and FEVDs

Description	Quick start	Menu	Syntax
Options	Remarks and examples	Stored results	Also see

Description

`irf ctable` makes a table or a combined table of IRF results. A table is made for specified combinations of named IRF results, impulse variables, response variables, and statistics. `irf ctable` combines these tables into one table, unless separate tables are requested.

`irf ctable` operates on the active IRF file; see [\[TS\] irf set](#).

Quick start

Combine tables of an orthogonalized IRF `myirf` and cumulative IRF `mycirf` for dependent variables `y1` and `y2`

```
irf ctable (myirf y1 y2 oirf) (mycirf y1 y2 cirf)
```

Same as above, but suppress confidence intervals and add a title

```
irf ctable (myirf y1 y2 oirf) (mycirf y1 y2 cirf), noci ///
title("My Title")
```

Note: `irf` commands can be used after `var`, `svar`, `ivsvar`, `vec`, `arma`, `arfima`, `lpirf`, `dsge`, `dsgenl`, or `xtvar`; see [\[TS\] var](#), [\[TS\] var svar](#), [\[TS\] var ivsvar](#), [\[TS\] vec](#), [\[TS\] arma](#), [\[TS\] arfima](#), [\[TS\] lpirf](#), [\[DSGE\] dsge](#), [\[DSGE\] dsgenl](#), or [\[XT\] xtvar](#).

Menu

Statistics > Postestimation

Syntax

```
irf ctable (spec1) [ (spec2) ... [ (specN) ] ] [ , options ]
```

where (*spec*_{*k*}) is

```
(irfname impulsevar responsevar stat [ , spec_options ])
```

irfname is the name of a set of IRF results in the active IRF file. *impulsevar* should be specified as an endogenous variable for all statistics except *dm* and *cdm*; for those, specify as an exogenous variable. *responsevar* is an endogenous variable name. *stat* is one or more statistics from the list below:

<i>stat</i>	Description
-------------	-------------

Main	
<i>irf</i>	impulse–response function
<i>oirf</i>	orthogonalized impulse–response function
<i>dm</i>	dynamic-multiplier function
<i>cirf</i>	cumulative impulse–response function
<i>coirf</i>	cumulative orthogonalized impulse–response function
<i>cdm</i>	cumulative dynamic-multiplier function
<i>fevd</i>	Cholesky forecast-error variance decomposition
<i>sirf</i>	structural impulse–response function
<i>sfevd</i>	structural forecast-error variance decomposition

<i>options</i>	Description
----------------	-------------

<i>set(filename)</i>	make <i>filename</i> active
<i>noci</i>	do not report confidence intervals
<i>stderror</i>	include standard errors for each statistic
<i>individual</i>	make an individual table for each combination
<i>title("text")</i>	use <i>text</i> as overall table title
<i>step(#)</i>	set common maximum step
<i>level(#)</i>	set confidence level; default is <i>level(95)</i>

collect is allowed; see [\[U\] 11.1.10 Prefix commands](#).

<i>spec_options</i>	Description
---------------------	-------------

<i>noci</i>	do not report confidence intervals
<i>stderror</i>	include standard errors for each statistic
<i>level(#)</i>	set confidence level; default is <i>level(95)</i>
<i>ititle("text")</i>	use <i>text</i> as individual subtitle for specific table

spec_options may be specified within a table specification, globally, or both. When specified in a table specification, the *spec_options* affect only the specification in which they are used. When supplied globally, the *spec_options* affect all table specifications. When specified in both places, options for the table specification take precedence. *ititle()* does not appear in the dialog box.

Options

`set(filename)` specifies the file to be made active; see [TS] [irf set](#). If `set()` is not specified, the active file is used.

`noci` suppresses reporting of the confidence intervals for each statistic. `noci` is assumed when the model was fit by `vec` because no confidence intervals were estimated.

`stderror` specifies that standard errors for each statistic also be included in the table.

`individual` places each block, or (*spec_k*), in its own table. By default, `irf ctable` combines all the blocks into one table.

`title("text")` specifies a title for the table or the set of tables.

`step(#)` specifies the maximum number of steps to use for all tables. By default, each table is constructed using all steps available.

`level(#)` specifies the default confidence level, as a percentage, for confidence intervals, when they are reported. The default is `level(95)` or as set by `set level`; see [U] [20.8 Specifying the width of confidence intervals](#).

The following option is available with `irf ctable` but is not shown in the dialog box:

`ititle("text")` specifies an individual subtitle for a specific table. `ititle()` may be specified only when the `individual` option is also specified.

Remarks and examples

[stata.com](http://www.stata.com)

If you have not read [TS] [irf](#), please do so.

Also see [TS] [irf table](#) for a slightly easier to use, but less powerful, table command.

`irf ctable` creates a series of tables from IRF results. The information enclosed within each set of parentheses,

```
(irfname impulsevar responsevar stat [ , spec_options ])
```

forms a request for a specific table.

The first part—*irfname impulsevar responsevar*—identifies a set of IRF estimates or a set of variance decomposition estimates. The next part—*stat*—specifies which statistics are to be included in the table. The last part—*spec_options*—includes the `noci`, `level()`, and `stderror` options, and places (or suppresses) additional columns in the table.

Each specific table displays the requested statistics corresponding to the specified combination of *irfname*, *impulsevar*, and *responsevar* over the step horizon. By default, all the individual tables are combined into one table. Also by default, all the steps, or periods, available are included in the table. You can use the `step()` option to impose a common maximum for all tables.

► Example 1

In [example 1](#) of [TS] [irf table](#), we fit a model using `var` and we saved the IRFs for two different orderings. The commands we used were

```
. use https://www.stata-press.com/data/r18/lutkepohl2
. var dln_inv dln_inc dln_consump
. irf set results4
. irf create ordera, step(8)
. irf create orderb, order(dln_inc dln_inv dln_consump) step(8)
```

We then formed the desired table by typing

```
. irf table oirf fevd, impulse(dln_inc) response(dln_consump) noci std
> title("Ordera versus orderb")
```

Using irf ctable, we can form the equivalent table by typing

```
. irf ctable (ordera dln_inc dln_consump oirf fevd)
> (orderb dln_inc dln_consump oirf fevd),
> noci std title("ordera versus orderb")
ordera versus orderb
```

Step	(1) oirf	(1) S.E.	(1) fevd	(1) S.E.
0	.005123	.000878	0	0
1	.001635	.000984	.288494	.077483
2	.002948	.000993	.294288	.073722
3	-.000221	.000662	.322454	.075562
4	.000811	.000586	.319227	.074063
5	.000462	.000333	.322579	.075019
6	.000044	.000275	.323552	.075371
7	.000151	.000162	.323383	.075314
8	.000091	.000114	.323499	.075386

Step	(2) oirf	(2) S.E.	(2) fevd	(2) S.E.
0	.005461	.000925	0	0
1	.001578	.000988	.327807	.08159
2	.003307	.001042	.328795	.077519
3	-.00019	.000676	.370775	.080604
4	.000846	.000617	.366896	.079019
5	.000491	.000349	.370399	.079941
6	.000069	.000292	.371487	.080323
7	.000158	.000172	.371315	.080287
8	.000096	.000122	.371438	.080366

- (1) irfname = ordera, impulse = dln_inc, and response = dln_consump.
- (2) irfname = orderb, impulse = dln_inc, and response = dln_consump.

The output is displayed in one table. Because the table did not fit horizontally, it automatically wrapped. At the bottom of the table is a list of keys that appear at the top of each column. The results in the table above indicate that the orthogonalized IRFs do not change by much. Because the estimated forecast-error variances do change, we might want to produce two tables that contain the estimated forecast-error variance decompositions and their 95% confidence intervals:

```
. irf ctable (ordera dln_inc dln_consump fevd)
> (orderb dln_inc dln_consump fevd), individual
```

Table 1

Step	(1) fevd	(1) Lower	(1) Upper
0	0	0	0
1	.288494	.13663	.440357
2	.294288	.149797	.43878
3	.322454	.174356	.470552
4	.319227	.174066	.464389
5	.322579	.175544	.469613
6	.323552	.175826	.471277
7	.323383	.17577	.470995
8	.323499	.175744	.471253

95% lower and upper bounds reported.

(1) irfname = ordera, impulse = dln_inc, and response = dln_consump.

Table 2

Step	(2) fevd	(2) Lower	(2) Upper
0	0	0	0
1	.327807	.167893	.487721
2	.328795	.17686	.48073
3	.370775	.212794	.528757
4	.366896	.212022	.52177
5	.370399	.213718	.52708
6	.371487	.214058	.528917
7	.371315	.213956	.528674
8	.371438	.213923	.528953

95% lower and upper bounds reported.

(2) irfname = orderb, impulse = dln_inc, and response = dln_consump.

Because we specified the `individual` option, the output contains two tables, one for each specific table command. At the bottom of each table is a list of the keys used in that table and a note indicating the level of the confidence intervals that we requested. The results from table 1 and table 2 indicate that each estimated function is well within the confidence interval of the other, so we conclude that the functions are not significantly different.

Stored results

`irf ctable` stores the following in `r()`:

Scalars

<code>r(ncols)</code>	number of columns in all tables
<code>r(k_umax)</code>	number of distinct keys
<code>r(k)</code>	number of specific table commands

Macros

<code>r(key#)</code>	#th key
<code>r(tnotes)</code>	list of keys applied to each column

Also see

[TS] **irf** — Create and analyze IRFs, dynamic-multiplier functions, and FEVDs

[TS] **var intro** — Introduction to vector autoregressive models

[TS] **vec intro** — Introduction to vector error-correction models

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).